

Examples 4,5 from %2.1 of the text, pages 83-84.

The Belgian demographer P.F. Verhulst introduced the logistic model around 1840, as a tool for studying human population growth. Our text demonstrates its superiority to the simple exponential growth model, and also illustrates why mathematical modelers must always exercise care, by comparing the two models to actual U.S. population data. here are actual U.S. populations from 1800-1990, see e.g. the table on page 84:

```
[ > restart: #clear Maple memory
> pops:= [[1800,5.3],[1810,7.2],[1820,9.6],[1830,12.9],
          [1840,17.1],[1850,23.2],[1860,31.4],[1870,38.6],
          [1880,50.2],[1890,63.0],[1900,76.2],[1910,92.2],
          [1920,106.0],[1930,123.2],[1940,132.2],[1950,151.3],
          [1960,179.3],[1970,203.3],[1980,225.6],[1990,248.7]]:
```

Unlike Verhulst, the book uses data from 1800, 1850 and 1900 to get constants in our two models. We let $t=0$ correspond to 1800.

Exponential Model: For the exponential growth model $P(t) = P_0 e^{(rt)}$ we use the 1800 and 1900 data to get values for P_0 and r :


```
[ > P0:=5.308;
  solve(P0*exp(r*100)=76.212,r);
          P0 := 5.308
          0.02664303814
> P1:=t->5.308*exp(.02664*t); #exponential model
          P1 := t -> 5.308 e(0.02664 t)
```

Logistic Model: We get P_0 from 1800, and use the 1850 and 1900 data to find k and M :

```
[ > P2:=t->M*P0/(P0+(M-P0)*exp(-M*k*t));
  #logistic function, with our P0
          P2 := t ->  $\frac{MP0}{P0 + (M - P0)e^{(-Mkt)}}$ 
> solve({P2(50)=23.192, P2(100)=76.212},{M,k});
          {k=0.0001677157274, M=188.1208275}
> M:=188.1208275;
  k:=.1677157274e-3;
  P2(t); #should be our logistic model function,
  #equation (11) page 82.
          M := 188.1208275
          k := 0.0001677157274
           $\frac{998.5453524}{5.308 + 182.8128275 e^{(-0.03155082142 t)}}$ 
```

Now compare the two models with the real data, and discuss:

```
> with(plots):  
  plot1:=plot(P1(t-1800),t=1800..1950,color=black, linestyle=3):  
    #this linestyle gives dashes for the exponential curve  
  plot2:=plot(P2(t-1800),t=1800..2000,color=black):  
  plot3:=pointplot(pops,symbol=cross):  
  display({plot1,plot2,plot3},title='U.S. population data  
and models');  
Warning, the name changecoords has been redefined
```



[>

The exponential model takes no account of the fact that the U.S. has only finite resources. Any ideas on why the logistic model begins to fail (with our parameters) around 1950?